Supplemental Digital Content 2:  SAS code.
*The following SAS code can be used to perform the analyses in this manuscript 
“Design and Analysis of Studies with Binary-Event Composite Endpoints: Guidelines for Anesthesia Research”, Anesthesia and Analgesia (2011).
*Disclaimer: this is general code for performing the analyses in the manuscript.
Some introductory and additional steps not included here are required to produce the results reported from Turan et al (2011) “Smoking and Perioperative Outcomes”, Anesthesiology. For example: reading the original NSQIP data, creating the exposure variable, propensity matching on baseline variables, assessment of mediation effects of smoking. As in Turan et al, all analyses were done adjusting for age, which was still slightly imbalanced between current and never smokers after propensity score matching. Complete code for either the Smoking study or Crystalloid-Colloid analyses are available from the first author upon request (maschae@ccf.org). The SAS dataset NSQIP_smoking_matched used below can be accessed in Supplemental Digital Content 1, http://links.lww.com/AA/A258;
*1. Datasets for analysis;

proc format;  value exp 1='1.  Current Smoker'   0='Never Smoker';
       value yesno  1='.Yes'  0= 'No';
%let components=  mort orgspcssi oupneumo reintub pulembol failwean cnscva cnscoma cdarrest cdmi othbleed othsysep othseshock;

*A.  1 observation per ID;

data  smoker_oneper;  set lib.NSQIP_smoking_matched;  
      *Set dataset with propensity-match current and  never smokers from NSQIP database;

    collapsed = max(of &components);  *collapsed composite outcome;

    count= sum(of &components);   *count outcome;

    format  &components  yesno.   smoker exp.;  run;

*B. 1 row per ID per component (for GEE analyses);

data Smokerlong;   set   Smoker_oneper;

    keep outcome  y  Smoker  id age;

    array yvec{*} &components;

    do outcome=1 to dim(yvec);

        y= yvec{outcome};

 
   output smokerlong;

    end; run;

*2. Individual component analysis (Section 2.2);

title "2. Individual component analysis adjusting for multiple comparisons and

         within-subject correlation ";

proc multtest data= Smoker_oneper   stepboot  nsample=200000  seed=12345;
      test ca(&components);    class Smoker;  run; 


*3. Collapsed composite analyis(Section 2.3.1);
title "3. Collapsed composite analyis";

proc logistic data=Smoker_oneper descending;

   class collapsed; 

   model  collapsed = Smoker age/rl; Run;

*4.  Count (Section 2.3.2);

title "4. Count: Wilcoxon rank sum test";   

proc npar1way data=Smoker_oneper  wilcoxon;

   var count; class Smoker; run;

title "4. Count: Proportional odds model";   

proc logistic data=Smoker_oneper  descending;

   model count= Smoker  age/rl;  Run;

*5.   Common effect GEE model (Section 3.1);

title "5. Common effect GEE analysis";

proc genmod data= Smokerlong   descending;

   class  id outcome;

   model y=outcome Smoker  age /  type3 dist=bin link=logit;

   repeated subject=id/ type=unstructured corrw; 

   estimate "common beta=0: estimate "  Smoker 1/e exp;  run;

*6: Distinct effects model tests various hypotheses by changing contrast L (Section 3.2)


1)
Treatment-outcome interaction


2)
Average effect test


3)
Average effect test with clinical importance weights;
title "6. Distinct effects models (Section 3.2)";
title 2 “6.1 Test for heterogeneity: assess the smoker x component interaction”;

proc genmod data= Smokerlong descending;

      class  id  outcome;

      model y= outcome  smoker Smoker*outcome  age/ type3 dist=bin link=logit covb;

      repeated subject=id/ type=un corrw; run;

title2 “6.2-3 Average relative effect tests”;                                                        
  


proc genmod data= Smokerlong descending;

    class  id  outcome;

       *remove intercept and main effect for treatment;

     model y= outcome  Smoker*outcome  age/ type3 dist=bin link=logit  noint covb;

     repeated subject=id/ type=un corrw;
%let L= .077 .077 .077 .077 .077 .077 .077 .077 .077 .077 .077 .077 .077;

     estimate "6.2. H0: avg beta=0 (wt=1/13=.077)"  Smoker*outcome  &L /e exp;

*Clinical importance weights.  Example: assign the first component, 30-day mortality, a weight double the other components (total of 13 components).   Since weights must sum to 1 for the average relative effect test, obtain weights by solve for X  in the equation  2X  +  12X  = 1.   Then X= 1/14=.071.  Mortality is thus assigned a weight of  2 (1/14) = 1/7 =0.143;

%let L= .143 .071 .071 .071 .071 .071 .071 .071 .071 .071 .071 .071 .071;

 estimate "6.3. H0: avg beta=0 (weighted)"  Smoker*outcome  &L /e exp;
*Can also test smoker-outcome interaction directly, using a 12 (i.e., K-1) row contrast as;
   contrast "1. Smoker-outcome interaction test"  
            Smoker*outcome   1 -1   0  0  0  0  0 0  0  0  0  0  0,

            Smoker*outcome   0  1  -1  0  0  0  0 0  0  0  0  0  0,
            Smoker*outcome   0  0   1 -1  0  0  0 0  0  0  0  0  0,

Smoker*outcome   0  0   0  1 -1  0  0 0  0  0  0  0  0,

            Smoker*outcome   0  0   0  0  1 -1 0  0  0  0  0  0  0,
            Smoker*outcome   0  0   0  0  0  1 -1 0  0  0  0  0  0,

Smoker*outcome   0  0   0  0  0  0  1 -1 0  0  0  0  0, 
Smoker*outcome   0  0   0  0  0  0  0  1 -1  0 0  0  0, 
Smoker*outcome   0  0   0  0  0  0  0  0  1 -1 0  0  0, 
Smoker*outcome   0  0   0  0  0  0  0  0  0  1 -1  0  0,

            Smoker*outcome   0  0   0  0 0  0   0  0  0  0  1 -1  0,
            Smoker*outcome   0  0   0  0 0  0   0  0  0  0  0  1 -1;
run;

*7.  Sample size calculations.  Although traditional sample size software is available for the more standard methods discussed, sample size planning for any composite outcome study is best done through simulations in which the researcher can specify the expected underlying effects, incidences and correlations for the various components. A comprehensive SAS power simulation macro incorporating all discussed methods is available by contacting the first author.  This program is appropriate for both randomized and non-randomized designs.;






